

Řešený příklad 1 z kap. 1.13 Klasifikace kvadrik

Příklad 1: Vyšetřete kvadriku $7x^2 + 6y^2 + 5z^2 - 4xy - 4yz - 22x + 24y + 2z + 30 = 0$.

```
[ > restart;  
[ > with(LinearAlgebra): with(linalg): with(plots):
```

Obecnou rovnici kvadriky můžeme zapsat a vytvořit užitím matice kvadriky K:

```
[ > X:=Vector[row]([x,y,z,1]);  
                                     X := [x, y, z, 1]  
[ > K:=Matrix(a,1..4,1..4,shape=symmetric);  
                                     K :=  $\begin{bmatrix} a(1,1) & a(1,2) & a(1,3) & a(1,4) \\ a(1,2) & a(2,2) & a(2,3) & a(2,4) \\ a(1,3) & a(2,3) & a(3,3) & a(3,4) \\ a(1,4) & a(2,4) & a(3,4) & a(4,4) \end{bmatrix}$   
[ > Kvadrika:=sort(expand(X.K.Transpose(X)),[x,y,z])=0;  
Kvadrika :=  $a(1,1)x^2 + 2a(1,2)xy + 2a(1,3)xz + a(2,2)y^2 + 2a(2,3)yz + a(3,3)z^2$   
            $+ 2a(1,4)x + 2a(2,4)y + 2a(3,4)z + a(4,4) = 0$ 
```

Příklad:

```
[ > RovKv:=7*x^2+6*y^2+5*z^2-4*x*y-4*y*z-22*x+24*y+2*z+30=0;  
RovKv :=  $7x^2 + 6y^2 + 5z^2 - 4xy - 4yz - 22x + 24y + 2z + 30 = 0$ 
```

Hodnoty koeficientů rovnice dané kvadriky, potřebné pro vytvoření její matice, získáme porovnáním obecného tvaru rovnice kvadriky "Kvadrika" s danou konkrétní rovnicí "RovKv". To vede na soustavu jednoduchých rovnic "SoustRovKoeff", z nichž každá má jako neznámou jeden z koeficientů $a(1,1)$, ..., $a(4,4)$ (viz následující řádky kódu).

Poznámka: Mohli jsme také určit každý koeficient zvlášť opakovaným použitím funkce `coeff` na rovnici "RovKv".

```
[ > SoustRovKoeff:={coeffs(lhs(collect(RovKv-Kvadrika,[x,y,z],distributed)))};  
SoustRovKoeff := {  $-2a(1,3), -4 - 2a(1,2), -4 - 2a(2,3), 5 - a(3,3), 30 - a(4,4),$   
                   $-a(1,1) + 7, -2a(1,4) - 22, -a(2,2) + 6, -2a(2,4) + 24, -2a(3,4) + 2$  }  
[ > KoeffRovKv:=solve(SoustRovKoeff,{a(1,1),a(1,2),a(1,3),a(1,4),a(2,2),  
a(3,3),a(2,3),a(2,4),a(3,4),a(4,4)});  
KoeffRovKv := {  $a(1,1) = 7, a(1,2) = -2, a(1,3) = 0, a(1,4) = -11, a(2,2) = 6, a(2,3) = -2,$   
                 $a(2,4) = 12, a(3,3) = 5, a(3,4) = 1, a(4,4) = 30$  }
```

Rovnosti "a(i,j)=ck" převedeme na přiřazovací příkazy užitím funkce assign.

```
[ > assign(KoefRovKv);
```

Tím se do obecného tvaru matice kvadriky K dosadí konkrétní hodnoty. Matice K dané kvadriky má pak tvar:

```
[ > K;
```

$$K := \begin{bmatrix} 7 & -2 & 0 & -11 \\ -2 & 6 & -2 & 12 \\ 0 & -2 & 5 & 1 \\ -11 & 12 & 1 & 30 \end{bmatrix}$$

Diskriminant kvadriky Δ :

```
[ > Delta:=det(K);
```

$$\Delta := -972$$

Diskriminant Δ je různý od nuly. Kvadrika je regulární.

Hlavní minor kvadriky A_{44} :

```
[ > SubK:=submatrix(K,1..3,1..3);
```

$$SubK := \begin{bmatrix} 7 & -2 & 0 \\ -2 & 6 & -2 \\ 0 & -2 & 5 \end{bmatrix}$$

```
[ > A44:=det(SubK);
```

$$A_{44} := 162$$

Hlavní minor kvadriky je různý od nuly. Kvadrika je středová.

Charakteristická rovnice kvadriky

Nejprve vytvoříme jednotkovou matici E:

```
[ > E:=evalm(array(1..3,1..3,identity));
```

$$E := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Potom definujeme matici ChM příslušné homogenní soustavy, která vede k charakteristické rovnici ChR:

```
[ > ChM:=evalm(SubK-lambda*E);
```

$$ChM := \begin{bmatrix} 7-\lambda & -2 & 0 \\ -2 & 6-\lambda & -2 \\ 0 & -2 & 5-\lambda \end{bmatrix}$$

```
[ > ChR:=det(ChM)=0;
```

$$ChR := 162 - 99\lambda + 18\lambda^2 - \lambda^3 = 0$$

Řešení charakteristické rovnice - vlastní čísla kvadriky:

```
[ > ResChR:=solve(ChR,lambda);
```

```
ResChR := 6, 3, 9
```

Pro snazší manipulaci můžeme vlastní čísla zapsat jako složky vektoru (uspořádané trojice) λ :

```
> lambda := [ResChR];  
  
lambda := [6, 3, 9]
```

Potom ke konkrétnímu vlastnímu číslu přistoupíme prostřednictvím odpovídajícího indexu (pořadového čísla v uspořádané trojici):

```
> lambda[1]; lambda[2]; lambda[3];  
  
6  
3  
9
```

Kanonický tvar rovnice kvadriky

Absolutní člen rovnice $\frac{\Delta}{A44}$:

```
> Delta/A44;  
  
-6  
> kr := lambda[1]*x^2+lambda[2]*y^2+9*z^2=-Delta/A44;  
  
kr := 6*x^2 + 3*y^2 + 9*z^2 = 6
```


Konečná podoba kanonického tvaru rovnice dané kvadriky:

```
> KanRovKv := kr / (abs(Delta/A44));  
  
KanRovKv := x^2 + y^2/2 + 3*z^2/2 = 1
```

Danou kvadrikou je trojosý elipsoid.

Délky poloos:

```
> a:=sqrt(1/coeff(lhs(KanRovKv),x,2));  
b:=sqrt(1/coeff(lhs(KanRovKv),y,2));  
c:=sqrt(1/coeff(lhs(KanRovKv),z,2));  
  
a = 1  
b = sqrt(2)  
c = sqrt(6)/3  
> plotsetup(inline,plotoptions=`portrait,noborder,shrinkby=0`);  
> implicitplot3d(KanRovKv,x=-2..2,y=-2..2,z=-2..2,grid=[30,30,30],  
style=patchcontour,orientation=[40,55],axes=normal,color=COLOR(R  
GB,250/255,250/255,250/255),light=[90,-5,1,1,1],tickmarks=[3,3,3  
]);
```



Vyšetření polohy kvadriky v původní soustavě souřadnic

Souřadnice středu kvadriky.

Uvažujme vzájemnou polohu přímky "Primka" a dané kvadriky "RovKv"

```

> Primka := [x=m+t*u, y=n+t*v, z=p+t*w];
          Primka := [x = m + t u, y = n + t v, z = p + t w]
> RovKv;
          7 x2 + 6 y2 + 5 z2 - 4 x y - 4 y z - 22 x + 24 y + 2 z + 30 = 0

```

Dosazení parametrických rovnic přímky za x, y, a z do rovnice kvadriky vede k následující rovnici s proměnnou t:

```

> RovKvPr := simplify(eval(RovKv, Primka));
RovKvPr := 7 m2 + 14 m t u + 7 t2 u2 + 6 n2 + 12 n t v + 6 t2 v2 + 5 p2 + 10 p t w + 5 t2 w2

```

$$-4mn - 4mtv - 4tun - 4t^2uv - 4np - 4ntw - 4tvp - 4t^2vw - 22m - 22tu + 24n + 24tv + 2p + 2tw + 30 = 0$$

Rovnici můžeme zapsat ve tvaru $A t^2 + B t + C = 0$, jejíž koeficienty A, B, C mají následující tvar:

```
> A:=coeff(lhs(RovKvPr),t^2);
      A := 7 u^2 + 6 v^2 + 5 w^2 - 4 u v - 4 v w
> B:=1/2*coeff(lhs(RovKvPr),t);
      B := 7 m u + 6 n v + 5 p w - 2 m v - 2 u n - 2 n w - 2 v p - 11 u + 12 v + w
> C:=coeff(lhs(RovKvPr),t,0);
      C := 7 m^2 + 6 n^2 + 5 p^2 - 4 m n - 4 n p - 22 m + 24 n + 2 p + 30
```

Středem kvadriky je bod $S = [m, n, p]$, pro jehož souřadnice je koeficient $B = 0$ bez ohledu na souřadnice $[u, v, w]$ směrového vektoru přímky.

```
> B1:=collect(B,[u,v,w]);
      B1 := (7 m - 11 - 2 n) u + (6 n - 2 m + 12 - 2 p) v + (-2 n + 5 p + 1) w
> RStr:=coeffs(B1,[u,v,w]);
      RStr := 6 n - 2 m + 12 - 2 p, 7 m - 11 - 2 n, -2 n + 5 p + 1
> RStr_res:=solve({RStr},{m,n,p});
      RStr_res := { m = 1, n = -2, p = -1 }
```

Střed kvadriky:

```
> S:=eval([m,n,p],RStr_res);
      S := [1, -2, -1]
```

Hlavní směry kvadriky

Řešíme příslušné homogenní soustavy rovnic, postupně pro všechny tři vlastní čísla.

1) $\lambda_1 = 3$

```
> MatHlSm1:=evalm(SubK-lambda[1]*E);
      MatHlSm1 := [ 1  -2  0
                   -2  0  -2
                   0  -2  -1 ]
> RovHlSm1:=geneqns(MatHlSm1,[u,v,w]);
      RovHlSm1 := { -2 u - 2 w = 0, u - 2 v = 0, -2 v - w = 0 }
> HlSm1:=solve(RovHlSm1,{u,v,w});
      HlSm1 := { u = 2 v, v = v, w = -2 v }
> u1:=eval([u,v,w],HlSm1);
      u1 := [2 v, v, -2 v]
```

Hlavní směr $u1$:

```
> u1:=eval(u1,v=1);
      u1 := [2, 1, -2]
```

2) $\lambda_2 = 6$

```
> MatH1Sm2:=evalm(SubK-lambda[2]*E);  
  
MatH1Sm2 :=  $\begin{bmatrix} 4 & -2 & 0 \\ -2 & 3 & -2 \\ 0 & -2 & 2 \end{bmatrix}$   
  
> RovH1Sm2:=geneqns(MatH1Sm2,[u,v,w]);  
RovH1Sm2 := { 4 u - 2 v = 0, -2 v + 2 w = 0, -2 u + 3 v - 2 w = 0 }  
> H1Sm2:=solve(RovH1Sm2,{u,v,w});  
H1Sm2 := { u = u, v = 2 u, w = 2 u }  
> u2:=eval([u,v,w],H1Sm2);  
u2 := [u, 2 u, 2 u]
```

Hlavní směr u_2 :

```
> u2:=eval(u2,u=1);  
u2 := [1, 2, 2]
```

3) $\lambda_3 = 9$

```
> MatH1Sm3:=evalm(SubK-lambda[3]*E);  
  
MatH1Sm3 :=  $\begin{bmatrix} -2 & -2 & 0 \\ -2 & -3 & -2 \\ 0 & -2 & -4 \end{bmatrix}$   
  
> RovH1Sm3:=geneqns(MatH1Sm3,[u,v,w]);  
RovH1Sm3 := { -2 u - 2 v = 0, -2 v - 4 w = 0, -2 u - 3 v - 2 w = 0 }  
> H1Sm3:=solve(RovH1Sm3,{u,v,w});  
H1Sm3 := { u = 2 w, v = -2 w, w = w }  
> u3:=eval([u,v,w],H1Sm3);  
u3 := [2 w, -2 w, w]
```

Hlavní směr u_3 :

```
> u3:=eval(u3,w=1);  
u3 := [2, -2, 1]
```

Odvodíme obecnou rovnici průměrové roviny sdružené se směrem $[u, v, w]$:

```
> U:=Vector[row]([u,v,w,0]);  
U := [u, v, w, 0]  
> PrumerR:=collect(expand(evalm(U*K&*Transpose(X))),[x,y,z])=0;  
PrumerR := (7 u - 2 v) x + (-2 u + 6 v - 2 w) y + (-2 v + 5 w) z - 11 u + 12 v + w = 0
```

Postupným dosazením hlavních směrů dostaneme obecné rovnice příslušných hlavních rovin:

```
> H1R1:=eval(PrumerR,[u=u1[1],v=u1[2],w=u1[3]]);  
H1R2:=eval(PrumerR,[u=u2[1],v=u2[2],w=u2[3]]);  
H1R3:=eval(PrumerR,[u=u3[1],v=u3[2],w=u3[3]]);  
H1R1 := 12 x + 6 y - 12 z - 12 = 0
```

$$HIR2 := 3x + 6y + 6z + 15 = 0$$

$$HIR3 := 18x - 18y + 9z - 45 = 0$$

Zobrazení kvadriky v původní poloze spolu s jejími osami a hlavními rovinami:

Můžeme si definovat barvu(y) pro obarvení grafu(ů):

```
[ > coll:=COLOR( RGB, 250/255, 250/255, 250/255 ) :
```

Graf kvadriky:

```
[ > Kvg:=implicitplot3d(RovKv,x=-2..3,y=-4..0,z=-3..2,axes=frame,color=coll,style=patchnogrid,grid=[40,40,40],light=[60,20,1,1,1],tickmarks=[3,3,3],orientation=[52,63],scaling=constrained):
[ > evalm(S+t*u1);
[ 1+2t, -2+t, -1-2t ]
```

Grafy jednotlivých os elipsoidu:


```
[ > o1g:=plot3d(evalm(S+t*u1),t=-2..2,j=-1..1,thickness=3):
[ > o2g:=plot3d(evalm(S+t*u2),t=-2..2,j=-1..1,thickness=3):
[ > o3g:=plot3d(evalm(S+t*u3),t=-2..2,j=-1..1,thickness=3):
```

Grafy jednotlivých hlavních rovin:

```
[ > H1R1g:=plot3d(solve(H1R1,z),x=-2..3,y=-4..0,color=blue,style=wireframe,contours=60):
[ > H1R2g:=plot3d(solve(H1R2,z),x=-2..3,y=-4..0,color=red,thickness=2,style=wireframe,contours=60):
[ > H1R3g:=plot3d(solve(H1R3,z),x=-2..3,y=-4..0,color=green,style=wireframe,contours=60):
```

K zobrazení více grafů v jedné soustavě použijeme příkaz plots[display]:

```
[ > display(Kvg,o1g,o2g,o3g,H1R1g,H1R2g,H1R3g,axes=frame,scaling=constrained,orientation=[143,75],view=[-2..3,-4..0,-3..2]);
```



- [>
- [>
- [>
- [>
- [>